Proof that, in the square multiplication table, the diagonal of squares describes a "ridge line" (imagine a roof), and as you "move off" the ridge, you go "downhill" (toward smaller numbers). So, for example, 900^2 = 810000, but moving off the ridge (perpendicularly), you get smaller numbers, so 901*899 =...

```
(%i2) 901*899;
```

(%o2) 809999

...a smaller number. Generalized proof follows. Pick any (x,y) pair of integers and move any distance e along the southwest-northeast axis (where (1,1) is the top-left corner of the table).

```
(%i3) x*y-(x-e)*(y+e);
```

(%o3) $x \cdot y - (x - e) \cdot (y + e)$

```
(%i4) expand(x*y-(x-e)*(y+e));
```

(%o4) $e \cdot y - e \cdot x + e^2$

Or...

```
(%i5) factorsum(x*y-(x-e)*(y+e));
```

(%o5) $e \cdot (y - x + e)$

If you think of positions BELOW the main diagonal and moving toward the southwest, so that e is a positive number and y is always greater than x, then you always get positive numbers (i.e., x decreses and y increases). The rest of this writeup assumes we're below that main diagonal. (The logic is the same for above the diagonal, obviously, although signs change.) And now, the d3 function I mentioned in the code comments:

```
(%i6) d3(n,m):=(836+n)*(836+m)-836^2;
```

(%o6) $\mathrm{d3}(n, m) := (836 + n) \cdot (836 + m) - 836^2$

```
(%i7) ratsimp(d3(n,m));
```

(%o7) $(m + 836) \cdot n + 836 \cdot m$

Find an expression that gives us m for a given n such that the delta will still be zero

```
(%i8) solve(d3(n,m)=0,m);
```

(%o8) $[m=-\frac{836\cdot n}{n+836}]$

Make the solution a function, "m for n".

```
(%i9)  m4(n):= 836*n/(n+836);
```

(%o9) $\mathrm{m4}\left(n\right):=\frac{836\cdot n}{n+836}$

Test. Back up one position on the x axis. (-1)

```
(%i10) m4(-1);
```

(%o10) $-\frac{836}{835}$

Need to shift y more than 1 (+1). Verify:

```
(%i11) d3(-1,1);
```

(%o11) -1

```
(%i12) d3(-1,2);
```

(%o12) 834

What's the biggest y-offset from 836 we can tolerate? 999-836.

```
(%i13) 999-836;
```

(%o13) 163

What x-offset will require that large a y-offset?

```
(%i14) solve(m4(n)=-163,n),numer;
```

rat: replaced 136.4044044044044 by 136268/999 = 136.4044044044044

(%o14) $[n=-136.4044044044044]$

Verify. Back x off by -136 and increase y the maximum (so... is (836-136)*(836+163) > 836^2 ?):

```
(%i15) d3(-136,163);
```

(%o15) 404

```
(%i16) d3(-137,163);
```

(%o16) -595

Here's another way to do it, that doesn't require an alternative function like m4:

```
(%i17) solve(d3(n,163)=0,n),numer;
```

rat: replaced 136.4044044044044 by 136268/999 = 136.4044044044044

(%o17) $[n = -136.4044044044044]$

So, how did I know the relationship between n and m isn't linear? For starters, the function doesn't have a linear form, but also, if you plot it, it doesn't look like a straight line.

```
(%i22) plot2d(m4(n),[n,-136,0]);
```

(%o22) [/Users/john/maxout.gnuplot_pipes,/Users/john/maxplot.qt] The real proof is in the derivative. Straight line ==> constant derivative.

```
(%i19) diff(m4(n),n);
```

(%o19) $\dfrac{836}{n+836} - \dfrac{836 \cdot n}{(n+836)^2}$

```
(%i20) ratsimp(diff(m4(n),n),n);
```

(%o20) $\dfrac{698896}{n^2 + 1672 \cdot n + 698896}$

That ain't a constant, so the m4 function isn't linear.

```
(%i23) plot2d(diff(m4(n),n),[n,-136,0]);
```

(%o23) [/Users/john/maxout.gnuplot_pipes,/Users/john/maxplot.qt] So, never mind trying to be too fancy. Just take all pairs in the triangle starting at (700,700) (which is 836 - 136) and find the biggest palindrome. Not super-elegant, but not TOO wasteful.